DISTRIBUČNÍ ÚLOHY

KONTEJNEROVÝ DOPRAVNÍ PROBLÉM, ROZŠÍŘENÁ ÚLOHA BATOHU (BIN PACKING PROBLEM), ÚLOHA OPTIMÁLNÍHO ROZMÍSTĚNÍ ZAŘÍZENÍ, ÚLOHA O POKRYTÍ.

POKRÝVACÍ A DĚLÍCÍ PROBLÉM (SET COVERING A SET PARTITIONING PROBLÉM)

Cílem je optimální pokrytí či dělení nějaké množiny. Mějme například úkoly $U = \{U_1, U_2 \dots U_m\}$, $M = \{1,2\dots m\}$. Dále mějme firmy $F_1, F_2 \dots F_n$. Každá firma zajišťuje určitou podmnožinu ze všech úkolů, a to za cenu c_j (tzn. každý úkol zajišťuje firma za stejnou cenu). V matici A máme informace o tom, zda je firma *j* schopna zajistit úkol *i*: a_{ij} = 1 tehdy, pokud je firma *j* schopna úkol *i* zajistit, jinak 0.

Cílem **pokrývacího problému** je vybrat firmy tak, aby byly co nejlevněji pokryty všechny úkoly.

$z = \sum_{j=1}^{n} c_j x_j \to min,$	*Minimalizujeme celkové náklady na zajištění úkolů. Často se minimalizuje pouze suma x_j, tedy počet použitých firem.*
$\sum_{j=1}^{n} a_{ij} x_j \geq 1, i = 1, 2, \dots, n,$	*a_{ij} = 1 tehdy, pokud je firma j schopná zajistit úkol i. Podmínka říká, že každý úkol musí zajišťovat **alespoň jedna** firma., a sčítáme vlastně jen přes ty firmy, které jsou schopny i-tý projekt zajistit, protože jinak je a_{ij} rovno 0. Případně při rozdělování pracovníků na směny (tzn. úkoly jsou směny) apod. nemusí být napravo nutně 1, někdy si přejeme, aby na směně byli minimálně dva pracovníci apod.*
$x_j \in \{0, 1\},\ j = 1, 2, \dots, n.$	*x_j = 1 tehdy, pokud je vybrána firma j*

Jednou z aplikací je například výstavba stanic rychlé pomoci v různých lokalitách (analogie firem), které mají obsluhovat určité obvody (analogie úkolů). V matici A je prvek a_{ij} roven 1 tehdy, pokud je stanice v *j*-té lokalitě ve stanovené dojezdové vzdálenosti od *i*-tého obvodu. c_{ij} pak značí například náklady na provoz stanice v *j*-té lokalitě.

$z = \sum_{j=1}^{n} c_j x_j \to min,$	*Minimalizujeme celkové náklady na provoz stanic.*
$\sum_{j=1}^{n} a_{ij} x_j \geq 1, i = 1, 2, \dots, n,$	*a_{ij} = 1 tehdy, pokud j-tá stanice v požadované dojezdové vzdálenosti. Každý obvod musí obsluhovat alespoň jedna stanice.*
$x_j \in \{0, 1\},\ j = 1, 2, \dots, n.$	*x_j = 1 pokud bude v lokalitě j postavena stanice*

Cílem **dělícího problému** je vybrat firmy tak, aby byly projekty rozděleny mezi ně tak, aby každý projekt zajišťovala právě jedna firma a náklady přitom byly minimální.

$z = \sum_{j=1}^{n} c_j x_j \to min,$	*Minimalizujeme celkové náklady na zajištění projektů. Často se minimalizuje pouze suma x_j, tedy počet použitých firem.*
$\sum_{j=1}^{n} a_{ij} x_j = 1, i = 1, 2, \dots, n,$	*a_{ij} = 1 tehdy, pokud je firma j schopná zajistit projekt i. Každý projekt musí zajišťovat **právě jedna** firma. Případně při rozdělování pracovníků na směny apod. nemusí být napravo nutně jednička, někdy si přejeme, aby na směně byli právě dva pracovníci apod.*
$x_j \in \{0, 1\},\ j = 1, 2, \dots, n.$	*x_j = 1 tehdy, pokud je vybrána firma j*

```
!V čajovně pracuje celkem 8 čajmenů a vrchní čajmen přemýšlí, kteří z nich by měli
tento týden přijít do práce. Každý večer od pondělí do pátku musí být v čajovně
alespoň dva z nich. Každý z nich sdělil vrchnímu čajmenovi, který večer by mohl
dorazit. Jednička označuje, že čajmen může přijít, nula že přijít nemůže.
```

	C1	C2	C3	C4	C5	C6	C7	C8
po	0	0	0	1	1	0	1	1
út	1	0	1	1	1	1	0	0
st	1	1	0	0	0	1	0	1
čt	0	1	0	1	1	0	1	1
pá	1	0	1	1	0	1	1	1

```
Cílem je minimalizovat počet čajmenů, kteří budou tento týden v práci.;
model:
sets:
cajmen/1..8/:x;
den/po,ut,st,ct,pa/;
prirazeni(den,cajmen): a;
endsets

data:
a =
0 0 0 1 1 0 1 1
1 0 1 1 1 1 0 0
1 1 0 0 0 1 0 1
0 1 0 1 1 0 1 1
1 0 1 1 0 1 1 1;
enddata

min = @sum(cajmen(j): x(j));
@for(den(i): @sum(cajmen(j): a(i,j)*x(j)) >=2);!pokrývací problém: aspoň 2 čajmeni
tam musí být každý večer;
!@for(den(i): @sum(cajmen(j): a(i,j)*x(j)) >=2);!dělící problém: právě 2 čajmeni;
@for(cajmen: @bin(x));
end
```

Výsledek pokrývacího problému: tento týden by měli přijít čajmeni číslo 1, 4 a 8.
Úloha má alternativní optimální řešení.

	C1	C2	C3	C4	C5	C6	C7	C8
po	0	0	0	1	1	0	1	1
út	1	0	1	1	1	1	0	0
st	1	1	0	0	0	1	0	1
čt	0	1	0	1	1	0	1	1
pá	1	0	1	1	0	1	1	1
x_j	1	0	0	1	0	0	0	1

ÚLOHA O POKRYTÍ

Typickým příkladem úlohy o pokrytí je situace, kdy máme *n* obvodů O_1, O_2…O_n, v nichž chceme postavit celkem *K* obslužných stanic, které všech těchto *n* obvodů obsluhovat, přičemž n > K. Navíc je třeba určit, které obvody budou obsluhovány kterou stanicí.

K dispozici máme informace o čase/dojezdové vzdálenosti ze stanice postavené v obvodu O_i do obvodu O_j a průměrnou frekvenci zásahů v každém obvodu. Cílem je rozhodnout, ve kterých obvodech postavit obslužné stanice a které obvody jim přiřadit tak, aby doba zásahu byla minimální.

Na rozdíl od přiřazovacího problému neplatí, že jeden prvek množiny musí být přiřazen právě jednomu jinému prvku, naopak jeden prvek množiny (jedna stanice) by měla obsluhovat více prvků.

Na rozdíl od pokrývacího problému tímto modelem určíme přesné přiřazení. U pokrývacího problému jsme pouze zjistili, které firmy vybrat, aby byly všechny úkoly splněny, ale nikoli která firma by měla dělat který úkol. Všechny úkoly zajišťovala daná firma s týmiž náklady. Tady se však náklady stanic na obsluhu jednotlivých obvodů (v podobě dojezdových vzdáleností) liší. Modelem zjistíme i jednoznačné přiřazení obvodů jednotlivým stanicím v závislosti na těchto nákladech.

Zavádí se dvě binární proměnné:

y_i, která je rovna 1, pokud je v lokalitě O_i zřízena stanice, jinak 0

x_{ij}, která se rovná 1, pokud stanice v obvodu O_i obsluhuje stanici v obvodě O_j

Náklady (čas či vzdálenost) na dojezd ze stanice v *i*-tém obvodě do *j*-tého obvodu označíme c_{ij}.
Frekvenci zásahů v *j*-tém obvodě označme f_j.

$z = \sum_{i=1}^{m} \sum_{j=1}^{n} c_{ij} x_{ij} f_j \to min$	*Minimalizujeme dojezdový čas/vzdálenost násobenou frekvencí zásahů.*
$\sum_{j=1}^{n} x_{ij} \le n y_i \quad i = 1, 2, \dots, m$	*(1) Když bude v obvodě i stanice, bude obsluhovat nejvýše n obvodů, jinak 0.*
$\sum_{i=1}^{m} x_{ij} = 1 \quad j = 1, 2, \dots, n$	*(2) Každý obvod musí být obsluhován právě jednou stanicí.*
$\sum_{i=1}^{m} y_i = K$	*(3) Celkem musí být zřízeno K stanic.*
$x_{ij} \in \{0, 1\}, \quad i = 1, 2 \dots m, j = 1, 2, \dots, n$	*(4) Je obvod j obsluhován stanicí i?*
$y_i \in \{0, 1\}, \quad i = 1, 2, \dots, m$	*(5) Je v obvodě i postavena stanice?*

Možné (ne nutně optimální) řešení by mohlo vypadat takto:

x_{ij}	O1	O2	O3	O4	O5	y_i
O1	4	3	7	8	3	1
O2	6	4	5	4	8	0
O3	8	9	7	2	9	0
O4	7	6	3	4	6	1
O5	6	6	7	6	7	0
f_j	10	8	5	12	9	

x_{ij}	O1	O2	O3	O4	O5	y_i	
O1	1	1	0	0	1	1	Σ =3
O2	0	0	0	0	0	0	Σ = 0
O3	0	0	0	0	0	0	
O4	0	0	1	1	0	1	
O5	0	0	0	0	0	0	
	Σ =1					Σ =2	

(1) Když je v i-tém obvodě stanice, obsluhuje nejvýše 5 obvodů. Jinak 0.

(5) Je v obvodě i postavena stanice?

(2) Každý obvod musí být obsluhován jednou stanicí.

(3) Celkem musí být postaveno K stanic.

(4) Obsluhuje stanice v i-tém obvodě j-tý obvod?

ÚLOHA OPTIMÁLNÍHO ROZMÍSTĚNÍ ZAŘÍZENÍ (PLANT LOCATION PROBLEM)

Máme k dispozici $M = \{1,2,\ldots m\}$ míst a v některých z nich chceme zřídit sklad. Každý sklad má kapacitu a_i a fixní náklady na jeho provoz jsou f_i. Dále máme N zákazníků = $\{1,2,\ldots n\}$ a každý z nich má určité požadavky b_j. Známe jednotkové přepravní náklady c_{ij} z i-tého skladu k j-tému zákazníkovi. Naším cílem je určit optimální rozmístění skladů tak, aby byly splněny požadavky zákazníků, a to s minimálními náklady.

Zavádí se jedna binární proměnná:

x_i, která je rovna 1, pokud je v místě Mi zřízen sklad, jinak 0

Dále se zavádí proměnná y_{ij}

y_{ij}, představuje množství zboží přepravovaného z i-tého skladu k j-tému zákazníkovi

$z = \sum_{i=1}^{m}\sum_{j=1}^{n} c_{ij}y_{ij} + \sum_{i=1}^{m} f_i x_i \to min$	*Minimalizujeme dojezdový čas/vzdálenost násobenou frekvencí zásahů.*
$\sum_{j=1}^{n} y_{ij} \le a_i x_i \quad i = 1,2 \ldots m$	*(1) Když bude v i-tém místě zřízen sklad, musí být součet přepraveného zboží do všech j míst menší než jeho kapacita.*
$\sum_{i=1}^{m} y_{ij} = b_j \quad j = 1,2,\ldots,n$	*(2) Musí být uspokojeny požadavky všech zákazníků.*
$y_{ij} \ge 0 \quad i = 1,2 \ldots m, j = 1,2,\ldots,n$	*(3) Kolik zboží přepravuje i-tý sklad k j-tému zákazníkovi?*
$x_i \in \{0,1\}, \quad i = 1,2,\ldots,n$	*(4) Je v místě i zřízen sklad?*

Jednodušší příklad:

```
!Mame k dispozici 6 mist, kde bychom mohli zridit sklad, a fixni naklady by byly: 
25, 15, 35, 30, 20 a 25 tisíc Kč.
Každý sklad má určitou kapacitu: 60, 50, 30, 35, 25 a 30 tisíc kusů.
Zároveň máme 5 zákazníků, kteří chtějí odebírat zboží z těchto skladů. Jejich 
požadavky jsou 12, 18, 23, 17 a 30 tisíc ks zboží.
Matice jednotkových přepravních nákladů je uvedena níže. Ve kterých místech máme 
postavit sklad a jak rozvrhnout rozvoz zboží?;
```

c_{ij}	O1	O2	O3	O4	O5	f_i	a_i
S1	2	3	2	4	5	25	60
S2	3	4	5	6	5	15	50
S3	1	1	7	8	2	35	30
S4	5	6	3	4	2	30	35
S5	5	6	4	3	2	20	25
S6	4	3	2	6	6	25	30
b_j	12	18	23	17	30		

```
model:
sets:
sklady/1..6/:f,x,kapacity; !f = fixni naklady, x = zrizen sklad?;
odberatele/1..5/:pozadavky;
preprava(sklady,odberatele): y,c; !c = prepravni naklady, y = objem prepravy;
endsets

min = @sum(preprava(i,j): y(i,j)*c(i,j)) + @sum(sklady(i): x(i)*f(i));
@for(odberatele(j): @sum(sklady(i): y(i,j)) = pozadavky(j));
@for(sklady: @bin(x));
@for(sklady(i): @sum(odberatele(j): y(i,j)) <= kapacity(i)*x(i)); !pokud bude v 
miste i zrizen sklad, pak nesmi preprava prekrocit pozadavky;
```

```
data:
kapacity = 60 50 30 35 25 30;
pozadavky = 12 18 23 17 30;
f = 25 15 35 30 20 25;
c =
2 3 2 4 5
3 4 5 6 5
1 1 7 8 2
5 6 3 4 2
5 6 4 3 2
4 3 2 6 6;

enddata
```

Řešení:

c_{ij}	O1	O2	O3	O4	O5	Σ	kap	xi
S1	5	0	23	17	0	45	60	1
S2	0	0	0	0	0	0	50	0
S3	7	18	0	0	5	30	30	1
S4	0	0	0	0	0	0	35	0
S5	0	0	0	0	25	20	25	1
S6	0	0	0	0	0	0	30	0
Σ	12	18	23	17	30			
poz	12	18	23	17	30			

Trošku složitější příklad:

Popis situace:

Společnost PHARMA objevila v Tichém oceánu tři ostrovy sopečného původu (O1, O2, O3), na nichž se vyskytuje zvláštní hornina, která by podle nich mohla být používaná pro lékařské účely. Proto se rozhodla zřídit na nich výzkumné stanice.

Problémem je, že kromě této horniny na ostrovech není nic jiného, proto musí PHARMA stanice zásobovat z pevniny. Vzhledem k tomu, že pevnina je velmi daleko, došla PHARMA na základě nákladové analýzy k závěru, že z ní nemá smysl dovážet do stanic vodu. Tu lze totiž na rozdíl od ostatních potřeb pro jejich provoz odebírat i ze čtyř neobydlených ostrovů poblíž (D1, D2, D3, D4).

Vodu je možné dopravovat v barelech, a to buď na malých lodích, nebo letecky. Barely mají kapacitu 1000 litrů vody a jsou přepravovány vždy plné. Do jedné lodi se vejde 60 barelů, letadlo může přepravit 40 barelů. Lodě ani letadla společnost kupovat nemusí, jelikož jich vlastní dostatek.

Náklady na přepravu jednoho letadla, respektive jedné lodě z ostrovů s vodou na ostrovy se stanicemi v tisících Kč jsou uvedeny v tabulce:

LOĎ	O1	O2	O3
D1	9	12	15
D2	18	15	6
D3	18	18	12
D4	15	21	15

LETADLO	O1	O2	O3
D1	12	16	20
D2	24	20	8
D3	24	24	16
D4	20	28	20

Na prvním ostrově (O1) je potřeba měsíčně 180 tisíc litrů vody, na druhém 120 tisíc litrů a na třetím 200 tisíc litrů.

Bylo zjištěno, že vzhledem k relativně malému průtoku vodních toků na dodavatelských ostrovech je možné z prvního ostrova (D1) měsíčně poskytnout 150 tisíc litrů vody, z druhého 100 tisíc litrů, z třetího 250 tisíc litrů a ze čtvrtého 200 tisíc litrů.

Aby bylo možné realizovat přepravu, je však nutné na dodavatelských ostrovech postavit buď malý přístav, nebo příletovou plochu (nazvěme trochu nadneseně „letiště“). Měsíční obsluha přístavu stojí 35 tisíc Kč, měsíční obsluha příletové plochy pak jen 18 tisíc Kč, na druhou stranu přeprava letadlem je dražší než přeprava lodí. Náklady na samotné zřízení přístavů a letišť jsou srovnatelné, a není tedy třeba je při optimalizaci uvažovat, protože vzhledem ke kapacitám dodavatelů je zřejmé, že budou postaveny celkem tři objekty (společnost by zřizovací náklady zahrnula do výpočtu v případě, že by výsledkem optimalizace byla stavba čtyř objektů, a zřizovací náklady by tudíž také hrály roli).

Protože má PHARMA obavu, že projekt bude stát velké množství peněz, informovala se u Evropské unie ohledně grantů. Zjistila, že EU je ochotná výzkum finančně podporovat, ale pouze ve výši 600 eur, tzn. 15 tisíc Kč měsíčně, a navíc jen za předpokladu, že se při výše zmíněné přepravě vody budou používat výhradně lodě, protože je to ekologičtější a nevzniká tolik emisí CO^2, proti kterým EU bojuje.

1. Problém:
2. Na kterých ostrovech s vodou se má zřídit přístav a na kterých letiště?
3. Z kterého ostrova na který je nejvýhodnější přepravovat vodu a jakým dopravním prostředkem?
4. V jakém množství má být voda přepravována, tzn. kolika loďmi či letadly?
5. Vyplatí se využít podpory EU?

Model 1:

```
!
Proměnné:
x = počet barelů s vodou, které jsou přepravované od D(i) k O(j)
y = počet lodí přepravovaných od D(i) k O(j)
w = počet letadel přepravovaných od D(i) k O(j)
P = přístav (binární proměnná, kdy P(i)=1, je-li na ostrově D(i) zřízen přístav,
P=0 jinak)
L = letiště (binární proměnná, kdy L(i)=1, je-li na ostrově D(i) zřízeno letiště,
P=0 jinak)

Parametry:
c = měsíční náklady na provoz přístavu
d = měsíční náklady na provoz letiště
naklod = náklady na přepravu jedné lodi od D(i) k O(j)
naklet = náklady na přepravu jednoho letadla od D(i) k O(j)
pozadavky = měsíční potřeba vody u jednotlivých odběratelů (uvedena v počtu barelů)
kapacity = množství, které je možno měsíčně dodat z jednotlivých ostrovů s vodou
(uvedeno v počtu barelů)
kaplod = kapacita lodi; kaplet = kapacita letadla;

model:
sets:
odberatele/1..3/:pozadavky;
dodavatele/1..4/:kapacity,P,L;
preprava(dodavatele,odberatele): naklod,naklet,x,y,w;
endsets

data:
naklod=
9      12     15
18     15     6
18     18     12
15     21     15;
naklet=
12     16     20
24     20     8
24     24     16
20     28     20;
pozadavky = 180 120 200;
kapacity = 150 100 250 200;
kaplod=60;
kaplet=40;
c=35;
d=18;
enddata

min = @sum(preprava: naklod*y)+@sum(preprava: naklet*w)+@sum(dodavatele:
c*P)+@sum(dodavatele:d*L);

@for(dodavatele: @bin(P));
@for(dodavatele: @bin(L));

@for(odberatele(j): @sum(dodavatele(i): x(i,j))=pozadavky(j));
@for(dodavatele(i): @sum(odberatele(j): x(i,j))<=kapacity(i)*(P(i)+L(i)));

@for(dodavatele(i): P(i)+L(i)<=1); !na ostrove muze byt jen jeden pristav ci
letiste
@for(preprava(i,j): x(i,j)<=kaplod*y(i,j)+kaplet*w(i,j));

@for(preprava: @gin(y));
@for(preprava: @gin(w));

@for(dodavatele(i): @sum(odberatele(j): w(i,j))<= 100*L(i));
@for(dodavatele(i): @sum(odberatele(j): y(i,j))<= 100*P(i));
!tyto dve podmínky zajistuji, ze pokud nebude na i-tém ostrove letiste, nebudou
z nej letat zadna letadla, totez v pripade lodi
end
```

Výsledky 1: Účelová funkce: 209 tisíc Kč

Matice X	O1	O2	O3	Celkem
D1	0	0	0	0
D2	0	0	80	80
D3	0	120	120	240
D4	180	0	0	180
Celkem	180	120	200	

Přístav	Letiště
0	0
0	1
1	0
1	0

Letadla	O1	O2	O3	Lodě	O1	O2	O3
D1	0	0	0	D1	0	0	0
D2	0	0	2	D2	0	0	0
D3	0	0	0	D3	0	2	2
D4	0	0	0	D4	3	0	0

Model 2

Přidáme podmínku:

```
@for(dodavatele(i): L(i) = 0);
```

Výsledky 2: Účelová funkce: 222 tisíc Kč

Matice X	O1	O2	O3	Celkem
D1	0	120	0	120
D2	0	0	0	0
D3	0	0	200	200
D4	180	0	0	180
Celkem	180	120	200	

Přístav	Letiště
1	0
0	0
1	0
1	0

Letadla	O1	O2	O3	Lodě	O1	O2	O3
D1	0	0	0	D1	0	2	0
D2	0	0	0	D2	0	0	0
D3	0	0	0	D3	0	0	4
D4	0	0	0	D4	3	0	0

Interpretace

Pokud by se společnost rozhodla nevyužít grant EU, pak by pro ni bylo nejvýhodnější zřídit na druhém ostrově letiště a na třetím a čtvrtém ostrově přístav. Z druhého dodavatelského ostrova poletí dvě letadla na třetí odběratelský ostrov, z třetího dodavatelského ostrova poplují dvě lodě na druhý a dvě na třetí odběratelský ostrov, ze čtvrtého dodavatelského ostrova poplují tři lodě na první odběratelský ostrov. Požadavky odběratelských ostrovů by byly splněny, kapacity dodavatelů by nebyly vyčerpány. Lodě i letadla by jely zcela plné a společnost by celkem zaplatila 209 tisíc Kč.

Pokud by se společnost rozhodla využít grantu EU, pak by nesměla vůbec stavět letiště. Proto by na prvním, třetím a čtvrtém dodavatelském ostrově zřídila přístavy. Z prvního dodavatelského ostrova by pluly dvě lodě na druhý odběratelský ostrov, ze třetího dodavatelského ostrova čtyři lodě na třetí odběratelský ostrov a ze čtvrtého dodavatelského ostrova tři lodě na první odběratelský. Kapacity dodavatelských ostrovů by rovněž nebyly překročeny. Čtvrtá loď plující z dodavatelského ostrova by nebyla zcela plná, ale vzhledem k nejistotě doby trvání celého projektu společnost neuvažuje při výpočtech využití této kapacity k vytváření zásoby na další měsíce. V případě zřízení tří přístavů by společnost měsíčně zaplatila 222 tisíc Kč. Jelikož od částky 222 tisíc Kč je možné odečíst 15 tisíc Kč, které společnost dostane od EU, celkové náklady při druhé alternativě budou jen 207 tisíc Kč, a proto se právě pro tuto alternativu společnost rozhodne.

ROZŠÍŘENÁ ÚLOHA BATOHU (BIN PACKING PROBLEM)

Uvažujme *n* typů výrobků, které označme V_1, V_2, …, V_n. U každého typu výrobku je dán jeho počet r_j, hmotnost a_j a objem b_j. K dispozici máme *m* kontejnerů o hmotnosti *K* a objemu *L.* Chtěli bychom umístit všechny tyto výrobky do kontejnerů tak, aby nebyla překročena jejich hmotnost ani objem, a aby byl zároveň počet použitých kontejnerů co nejmenší.

Zavádí se jedna binární proměnná:

x_i, která je rovna 1, pokud je *i*-tý kontejner obsazen, jinak 0.

Dále se zavádí nezáporná celá proměnná y_{ij}

y_{ij}, představuje množství *j*-tého výrobku v *i*-tém kontejneru

$z = \sum_{i=1}^{m} x_i \to min$	*Minimalizujeme počet použitých kontejnerů.*
$\sum_{i=1}^{m} y_{ij} = r_j \quad j = 1,2...n$	*(1) Pro každý výrobek typu j musí platit, že do všech kontejnerů dohromady musíme umístit všech r_j výrobků.*
$\sum_{j=1}^{n} a_j y_{ij} \le K x_i \quad i = 1,2...m$	*(2) Pro každý kontejner i musí platit, že jeho kapacita není překročena (hmotnost).*
$\sum_{j=1}^{n} b_j y_{ij} \le L x_i \quad i = 1,2...m$	*(3) Pro každý kontejner i musí platit, že jeho kapacita není překročena (objem).*
$y_{ij} \ge 0, celé \quad i = 1,2 \dots m, j = 1, 2, \dots, n$	*(4) Kolik výrobku j je v kontejneru i?*
$x_i \in \{0, 1\}, \quad i = 1, 2, \dots, n$	*(5) Je i-tý kontejner obsazen?*

```
!Firma má možnost si pronajmout 10 identických kontejnerů o kapacitě 5000 kg,
s jejichž pomocí má přepravit 3 druhy výrobků v následujícím množství: 65 ks, 68 ks
a 150 ks.
Hmotnost výrobků je 50 kg, 86 kg a 63 kg. Cílem je minimalizovat náklady spojené s
pronájmem kontejnerů, tedy vlastně počet použitých kontejnerů.;

model:
sets:
vyrobek/1..3/:pozadavek,hmotnost;
kontejner/1..10/:x;
preprava(kontejner,vyrobek): y;
endsets

data:
hmotnost = 40 86 63;
pozadavek = 65 68 150;
kapacita = 5000;
enddata

min = @sum(kontejner: x);
@for(vyrobek(j): @sum(kontejner(i): y(i,j)) = pozadavek(j));
@for(kontejner: @bin(x));
@for(kontejner(i): @sum(vyrobek(j): y(i,j)*hmotnost(j)) <=x(i)*kapacita);
end
!Řešením jsou 4 použité kontejnery.;

Zdroj: Ing. Jan Fábry, Ph.D.: 4EK314 Diskrétní modely
```

KONTEJNEROVÝ DOPRAVNÍ PROBLÉM

Kontejnerový dopravní problém je modifikací klasického dopravního problému. V něm je cílem rozvézt zboží od dodavatelů $D_1, D_2 \dots D_m$ k odběratelům $O_1, O_2 \dots O_n$ tak, aby nebyla překročena kapacita dodavatelů a_i, byly uspokojeny požadavky odběratelů a celkové náklady na přepravu byly minimální.

$z = \sum_{i=1}^{m}\sum_{j=1}^{n} c_{ij}x_{ij} \rightarrow min$	*Minimalizujeme náklady na přepravu, kde c_{ij} jsou jjednotkové náklady na přepravu od dodavatele i k odběrateli ija x_{ij} je objem přepravy.*
$\sum_{j=1}^{n} x_{ij} \leq a_i \quad i = 1,2 \dots m$	*(1) Pro každého dodavatele musí platit, že suma zboží přepravená od něj k odběratelům nepřekročí jeho kapacitu.*
$\sum_{i=1}^{m} x_{ij} = b_j \quad j = 1,2 \dots n$	*(2) Pro každého odběratele musí platit, že suma zboží přepravená k němu od dodavatelů se bude rovnat jeho požadavkům.*
$x_{ij} \geq 0 \qquad i = 1,2 \dots m, j = 1, 2, \dots, n$	*(3) Kolik výrobků je přepravováno z i do j?*

Na rozdíl od dopravního problému se u kontejnerového dopravního problému přeprava realizuje pomocí kontejnerů o kapacitě *K*. Uvažujme dodavatele $D_1, D_2 \dots D_m$ s kapacitou a_i a odběratele $O_1, O_2 \dots O_n$ s požadavky b_j. Známe přepravní náklady na jeden kontejner od *i*-tého dodavatele k *j*-tému odběrateli. Naším cílem je minimalizovat celkové přepravní náklady tak, abychom nepřekročili kapacitu kontejneru ani kapacitu jednotlivých dodavatelů a zároveň uspokojili požadavky odběratelů.

Zavádí se dvě proměnné:

- x_{ij}, představuje objem přepravy od dodavatele *i* k odběrateli *j*
- y_{ij}, představuje počet kontejnerů přepravených od dodavatele *i* k odběrateli *j* (celočíselná)

$z = \sum_{i=1}^{m}\sum_{j=1}^{n} c_{ij}y_{ij} \rightarrow min$	*Minimalizujeme náklady na přepravu kontejnerů.*
$x_{ij} \leq Ky_{ij} \quad i = 1,2 \dots m, j = 1,2 \dots n$	*(1) Pokud bude přepravován kontejner z i do j, pak v něm musí být přepravováno nejvýše tolik zboží, kolik činí jeho kapacita.*
$\sum_{j=1}^{n} x_{ij} \leq a_i \quad i = 1,2 \dots m$	*(2) Pro každého dodavatele musí platit, že suma zboží přepravená od něj k odběratelům nepřekročí jeho kapacitu.*
$\sum_{i=1}^{m} x_{ij} = b_j \quad j = 1,2 \dots n$	*(3) Pro každého odběratele musí platit, že suma zboží přepravená k němu od dodavatelů se bude rovnat jeho požadavkům.*
$y_{ij} \geq 0, celé \qquad i = 1,2 \dots m, j = 1, 2, \dots, n$	*(4) Kolik kontejnerů je přepravováno z i do j?*
$x_{ij} \geq 0 \qquad i = 1,2 \dots m, j = 1, 2, \dots, n$	*(5) Kolik výrobků je přepravováno z i do j?*

```
!Máme 4 dodavatele s kapacitami 30, 25, 21 a 35 tun a 5 odběratelů s požadavky
15,17, 22, 12 a 33 tun. Přeprava probíhá v kontejnerech o kapacitě 10 tun. Náklady
spojené s přepravou 1 kontejneru od jednotlivých dodavatelů k odběratelům jsou dány
maticí C. Úkolem je splnit požadavky odběratelů a minimalizovat celkové přepravní
náklady
```

C =	6	2	6	7	9
	4	9	5	3	6
	8	8	1	5	3
	7	4	5	8	5

```
;
model:
sets:
dodavatele/1..4/:kapacity;
odberatele/1..5/:pozadavky;
preprava(dodavatele,odberatele):x,y,C;
endsets

data:
kapacity = 30 25 21 35;
pozadavky = 15 17 22 12 33;
C = 6 2 6 7 9 4 9 5 3 6 8 8 1 5 3 7 4 5 8 5;
KAP = 10;
enddata

min = @sum(preprava: y*C);
@for(odberatele(j): @sum(dodavatele(i): x(i,j)) = pozadavky(j));
@for(dodavatele(i): @sum(odberatele(j): x(i,j)) <=kapacity(i));
@for(preprava: x<=KAP*y);
@for(preprava:@gin(y));
end
Zdroj: Ing. Jan Fábry, Ph.D.: 4EK314 Diskrétní modely
```

ZDROJE:

Ing. J. Fábry, Ph.D.: přednášky 4EK314 Diskrétní modely, 2011.